



# Courbe de Von Koch

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

Dans le cadre des séries géométriques, cette feuille Maple a été développée pour les illustrations d'un exercice de mon cahier de notes pour le cours EED. La présente feuille est une version documentée. La lecture de cette feuille de travail est l'occasion pour le lecteur d'aborder la récursivité dans la programmation (pour la courbe de Von Koch) et aussi pour constater la possibilité d'inclure l'appel à une procédure locale appelée dans une procédure (flocon de Von Koch).

Bonne lecture à tous !

\* Ce document Maple est exécutable avec la version 2020.1


La géométrie fractale permet de caractériser des objets ayant une forme irrégulière possédant la propriété d'invariance par changement d'échelle. De manière simplifiée, on définit un objet fractal comme un objet ayant une homothétie interne, c'est-à-dire qu'une portion de l'objet est identique à l'objet complet. Si vous regardez un objet fractal en le grossissant régulièrement, vous allez voir encore la même forme. On dit d'un objet fractal qu'il possède la particularité d'autosimilarité.

Le terme "fractal" vient du latin, "fractus" qui désigne un objet fracturé, de forme très irrégulière. C'est Benoît Mandelbrot qui, en 1975, a introduit ce terme pour désigner ces fameux objets mathématiques.

La courbe de Von Koch (1870-1924) est un exemple de fractal. Cette courbe a été publiée en 1904 et le titre de l'article mérite d'être cité : « Sur une courbe continue sans tangente, obtenue par une construction géométrique élémentaire ».

Cette courbe peut être dessinée en réalisant l'algorithme suivant:

- Étape 1: tracer un segment de longueur  $a$ ;
- Étape 2: au tiers milieu sur ce segment, construire vers l'extérieur, le triangle équilatéral de côté  $\frac{a}{3}$  et supprimer la base de ce triangle;
- Étape  $k$ ,  $k = 3, 4, \dots$  : répéter indéfiniment cette construction sur chaque segment ainsi formés.



## Initialisation


```
> restart;  
with(plots, setoptions):  
setoptions(size=[300,300]);
```



## Courbe de Von Koch

```
> Courbe_Von_Koch:=proc(n::posint)
  local k,C,Von_Koch,Opts;
  Opts:=[args[2..nargs]];
  Von_Koch:=proc(n::posint,xa,ya,xb,yb)
    option remember;
    local x1,y1,x2,y2,x3,y3;
    if n=1 then [xa,ya],[xb,yb]
    else
      x1:=evalf((2*xa+xb)/3);y1:=evalf((2*ya+yb)/3);
      x3:=evalf((xa+2*xb)/3);y3:=evalf((ya+2*yb)/3);
      x2:=evalf(1/2*(x1+x3-sqrt(3)*(y3-y1)));
      y2:=evalf(1/2*(y1+y3+sqrt(3)*(x3-x1)));
      Von_Koch(n-1,xa,ya,x1,y1),Von_Koch(n-1,x1,y1,x2,y2),Von_Koch(n-1,
x2,y2,x3,y3),Von_Koch(n-1,x3,y3,xb,yb);
    end if
  end proc;
  C:=plot([Von_Koch(n,0,0,1,0)],op(Opts));
  plots[display](C,axes=none,scaling=constrained,op(Opts))
end proc;
```

Voici les cinq premières étapes de la construction de la courbe de Von Koch.

```
> seq(print(Courbe_Von_Koch(k,view=[0..1,-.3..0.3])),k=1..5);
```






Les calculs nécessaires pour la construction de cette courbe croissent de manière astronomique. L'affichage des cinq ou six premières étapes est assez rapide mais à partir de la neuvième étape, là ça ralenti. Inutile de vouloir afficher l'étape 12 et plus car le temps de calcul est très, très long et même si on persévère, le nombre de points calculés étant astronomique, l'utilisation de la loupe pour l'agrandissement est, à toute fin pratique, peu utile.

Voyez par vous-même. L'affichage de l'étape 10 est assez rapide mais essayez la loupe sur ce tracé.

```
> Courbe_Von_Koch(10,color=navy);
```



Le flocon de Von Koch est la courbe que l'on obtient en prenant un triangle équilatéral de côté  $a$  comme initiateur au lieu d'un segment de longueur  $a$ . Sur chaque segment, nous appliquons l'algorithme de la courbe de Von Koch.




Étape 1




Étape 2




Étape 3



Étape 4



Étape 5



Étape 6




## Flocon de Von Koch



```
> Flocon_Von_Koch:=proc(n::posint)
  local k,C1,C2,C3,Opts,Von_Koch;
  Opts:=[args[2..nargs]];
  Von_Koch:=proc(n::nonnegint,xa,ya,xb,yb)
    option remember;
    local x1,y1,x2,y2,x3,y3;
    if n=1 then [xa,ya],[xb,yb]
    else
      x1:=evalf((2*xa+xb)/3);y1:=evalf((2*ya+yb)/3);
      x3:=evalf((xa+2*xb)/3);y3:=evalf((ya+2*yb)/3);
      x2:=evalf(1/2*(x1+x3-sqrt(3)*(y3-y1)));
      y2:=evalf(1/2*(y1+y3+sqrt(3)*(x3-x1)));
      Von_Koch(n-1,xa,ya,x1,y1),Von_Koch(n-1,x1,y1,x2,y2),Von_Koch(n-1,
x2,y2,x3,y3),Von_Koch(n-1,x3,y3,xb,yb);
    end if
  end proc;
  C1:=plot([Von_Koch(n,0,0,1/2,sqrt(3)/2)],op(Opts));
```

```
C2:=plot([Von_Koch(n,1/2,sqrt(3)/2,1,0)],op(Opts)):
C3:=plottools[rotate](plot([Von_Koch(n,0,0,1,0)],op(Opts)),-Pi,[1/2,
0]):
plots[display]([C1,C2,C3],axes=none,scaling=constrained,op(Opts))
end proc:
```

Voici les cinq premières étapes de la construction de la courbe de Von Kock.


```
> seq(print(Flocon_Von_Koch(k,view=[0..1,-0.35..1])),k=1..5);
```





Avec la loupe, explorez l'autosimilarité de l'étape 8 ci-dessous du flocon de Von Koch.

```
> Flocon_Von_Koch(8,color=magenta);
```



## Exercices

No. 1 *Courbe de Von Vock*

- Calculer  $L_k$ , la longueur de la courbe de Von Kock après  $k$  étapes (la longueur initiale du segment est  $a$ ).
- Déterminer la longueur  $L_k$ , lorsque  $k \rightarrow \infty$ .

No. 2 *Flocon de Von Kock*

- Obtenir le périmètre  $P_k$  du flocon de Von Koch après  $k$  étapes (la longueur initiale du côté du triangle équilatéral est  $a$ ).
- Déterminer la longueur  $P_k$ , lorsque  $k \rightarrow \infty$ .
- Calculer l'aire  $S_k$  du flocon de Von Koch après  $k$  étapes.
- Déterminer la longueur

$S_k$ , lorsque  $k \rightarrow \infty$ .

e) Quelle particularité possède le flocon de Von Koch au regard de son périmètre et de son aire lorsque  $k \rightarrow \infty$ .